
===================A_Simple_Way_To_Create_Digital_Randomness==============================

1) A random signal can be generated as an array which can be applied to a voltage
   source as a piece wise linear input.
2) One use for this is to generate a stream of completly random ones and zeros.

===================MacSpiceCode====================================================
Simple_Digital_Randomness
*================Need_A_voltage_Source_to_alter====================="
V1               V1      0      dc    0
Vref             VR      0      dc    0  PULSE( .25  .75 1n 1n  1n  1u 2u )
.control
set              pensize = 2
echo             "===================Want_50_1us_periods==============================="
let n =          50
let tstep =      1us
let period_t =   n*tstep
echo             "Sample_Period_s =  $&tstep"
echo             "===================Create_arrays_=============="
unlet pwl_1
unlet noise
unlet ii
let pwl_1 =      vector(4*n)
let noise =      vector(n)
let ii =         vector(1*$&n)
echo             "===================create_Noise_array====================="
let index =      0
repeat           $&n
let              noise[index] = pos(rnd(127)-64)
let index =      index + 1
end
*plot            noise  vs ii
echo             "===================create_PWL_array====================="
                 pwl_1[0] = 0
                 pwl_1[1] = noise[0]
                 pwl_1[2] = tstep -1n
                 pwl_1[3] = noise[0]
let  n2 =        n-1
let index =      1
repeat           $&n2
                 pwl_1[0+4*index] = pwl_1[4*index-4] +tstep
                 pwl_1[1+4*index] = noise[index]
                 pwl_1[2+4*index] = pwl_1[0+4*index] +tstep -1n
                 pwl_1[3+4*index] = noise[index]
let index =      index + 1
end
echo             "===================Install_the_PWL_array============================"
alter            @v1[pwl] = pwl_1
echo             "===================Run_and_Plot====================================="
let period_s =   tstep/10
*tran            1ns 100us
tran             $&period_s $&period_t   0  $&period_s
plot             v1  vr 
.endc
.end

